# Maintainer: AndyRTR <andyrtr@archlinux.org>
# Maintainer: Jan de Groot <jgc@archlinux.org>

pkgname=inputproto
pkgver=2.3.2
pkgrel=1
pkgdesc="X11 Input extension wire protocol"
arch=(any)
license=('custom')
url="https://xorg.freedesktop.org/"
makedepends=('xorg-util-macros' 'asciidoc')
source=(${url}/releases/individual/proto/${pkgname}-${pkgver}.tar.bz2{,.sig})
sha256sums=('893a6af55733262058a27b38eeb1edc733669f01d404e8581b167f03c03ef31d'
            'SKIP')
validpgpkeys=('3C2C43D9447D5938EF4551EBE23B7E70B467F0BF')

build() {
  cd ${pkgname}-${pkgver}
  ./configure --prefix=/usr
  make
}

package() {
  cd ${pkgname}-${pkgver}
  make DESTDIR="${pkgdir}" install

  install -m755 -d "${pkgdir}/usr/share/licenses/${pkgname}"
  install -m644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/"
}
